Automatyzacja Accessa

Sub PrzykladAccess()

' Pamiętaj aby zaznaczyć odwołania do biblioteki Access'a w Tools|References

    Dim appAccess As Access.Application

    ' Pobranie odwołania do obiektu aplikacji Access'a

    Set appAccess = New Access.Application

    With appAccess

        ' Otwórz przykładową bazę danych

        ' Jeśli potrzeba zmień odpowiednio ścieżkę dostępu do pliku

        .OpenCurrentDatabase "Northwind"

        ' Otwórz formularz

        .DoCmd.OpenForm "Klienci", acNormal

        ' Pokaż okno aplikacji

        .Visible = True

        ' Zamknij bazę danych

        .CloseCurrentDatabase

        ' Zakończ aplikację

        .Quit

    End With

    ' Zamknij odwołania do obiektu

    Set appAccess = Nothing

End Sub


Automatyzacja Excela

Sub PrzykladExcel()

' Pamiętaj aby zaznaczyć odwołania do bibliotek Excela

    Dim appExcel As Excel.Application

    Dim wbk As Workbook

    ' Ustalenie odwołania do obiektu aplikacji Excela

    Set appExcel = New Excel.Application

    appExcel.Visible = True

    Set wbk = appExcel.Workbooks.Add

    With wbk

        ' Wypełnij komórki arkusza od A1 do A5 wartością 5

        .ActiveSheet.Range("A1:A5") = "5"

        ' W komórce A6 podsumuj kolumnę

        .ActiveSheet.Range("A6").Formula = "=Sum(A1:A5)"

        ' Zachowaj arkusz na dysku.

        ' Jeśli arkusz o takiej nazwie już istnieje Excel poprosi o potwierdzenie operacji

        .SaveAs "test.xls"

        ' Zamknij arkusz

        .Close

    End With

    ' Zamknij Excela

    appExcel.Quit

    ' Wyczyść odwołania do obiektu

    Set wbk = Nothing

    Set appExcel = Nothing

End Sub

Automatyzacja Outlooka

Sub PrzykladOutlook()

' Pamiętaj aby zaznaczyć odwołania do bibliotek Outlooka

    Dim appOutlook As Outlook.Application

    Dim eml As MailItem

    ' Ustalenie odwołania do obiektu aplikacji Outlooka

    Set appOutlook = New Outlook.Application

    ' Utworzenie obiektu listu elektronicznego

    Set eml = appOutlook.CreateItem(olMailItem)

    eml.Display

    With eml

        ' Dodaj temat, treść oraz adresata listu

        .Subject = "Temat listu."

        .Body = "Treść listu."

        .Recipients.Add ("test@chip.pl")

        ' Wyślij list

        .Send

    End With

    ' Wyczyść odwołania do obiektu

    Set eml = Nothing

    Set appOutlook = Nothing

End Sub    

Automatyzacja PowerPointa

Sub PrzykladPowerPoint()

' Pamiętaj aby zaznaczyć odwołania do bibliotek Outlooka

    Dim appPowerPoint As PowerPoint.Application

    ' Ustalenie odwołania do obiektu aplikacji PowerPointa

    Set appPowerPoint = New PowerPoint.Application

    appPowerPoint.Visible = True

    ' Otwórz przykładową prezentację

    appPowerPoint.Presentations.Open "przyklad.ppt"

    ' Pokaż komunikat

    MsgBox "W tym miejscu, o ile plik z prezentacją istnieje, PowerPoint otworzy i wyświetli prezentację."

    ' Zamknij prezentację

    appPowerPoint.ActivePresentation.Close

    ' Zamknij PowerPointa

    appPowerPoint.Quit

    ' Wyczyść odwołania do obiektu

    Set appPowerPoint = Nothing

End Sub

Automatyzacja Worda

Sub PrzykladWord()

' Pamiętaj aby zaznaczyć odwołania do bibliotek Worda

    Dim appWord As Word.Application

    ' Ustalenie odwołania do obiektu aplikacji Worda

    Set appWord = New Word.Application

    With appWord

        ' Utwórz nowy dokument

        .Documents.Add

        .Visible = True

        ' Wpisz przykładowy tekst

        .Selection.TypeText ("Drabina z powyłamywanymi szczeblami.")

        ' Zapisz i zamknij dokument

        .ActiveDocument.SaveAs "test1.doc"

        .ActiveDocument.Close

        ' Zamknij Worda

        .Quit

    End With

    ' Wyczyść odwołania do obiektu

    Set appWord = Nothing

End Sub

